Разложение на элементарные дроби и интегрирование рациональных выражений

М.Д. Малых, РУДН

17 ноября 2022 г.

Содержание

1. Поле $\mathbb{Q}$

Всякое рациональное число можно представить как отношение двух целых чисел. Условимся считать их положительными, а знак писать впереди дроби, при необходимости относя его к числителю.

Определение 1. Дробь называется правильной, если ее числитель по модулю меньше знаменателя.

Определение 2. Правильная дробь называется простой (partial fraction), если ее знаменатель является степенью простого числа, а числитель меньше этого простого числа.

Теорема 1. Всякое рациональное число можно представить как сумму целого числа и нескольких простых дробей.

Доказательство. Пусть дана дробь f/g. Разложим знаменатель на множители

$$g = p_1^{m_1} \dots p_s^{m_s}.$$

Рассмотрим разность

$$\frac{f}{g} - \frac{c_1}{p_1^{m_1}} = \frac{f - c_1 p_2^{m_2} \dots p_s^{m_s}}{p_1^{m_1} \dots p_s^{m_s}}$$

и подберем c_1 так, чтобы

$$f - c_1 p_2^{m_2} \dots p_s^{m_s} = 0 \mod p_1.$$

Поскольку $\mathbb{Z}/(p_1) = \mathrm{GF}(p_1)$ — поле и

$$p_2^{m_2} \dots p_s^{m_s} \neq 0 \mod p_1,$$

это уравнение имеет решение относительно c_1, причем $0 \leq c_1 < p_1$. Но тогда

$$\frac{f}{g} - \frac{c_1}{p_1^{m_1}} = \frac{f_1}{p_1^{m_1-1} \dots p_s^{m_s}}$$

Рассмотрев дробь

$$\frac{f_1}{p_1^{m_1-1} \dots p_s^{m_s}}$$

вместо исходной, мы найдем c_2, такое, что

$$\frac{f_1}{p_1^{m_1-1} \dots p_s^{m_s}} - \frac{c_2}{p_1^{m_1-1}} = \frac{f_2}{p_1^{m_1-2} \dots p_s^{m_s}}$$

Действия так далее, мы сначала уберем множитель p_1, потом p_2 и так до полного исчезновения знаменателя. В итоге мы получим, что f/g без простых дробей

$$\frac{c_1}{p_1^{m_1}}, \quad \frac{c_2}{p_1^{m_1-1}}, \dots$$

является целым числом. □

Определение 3. Представление дроби в виде сумму целого числа и нескольких простых дробей называется разложением на простые дроби (partial fraction dicomposition).

Пример 1. Для дроби 113/40 имеем:

```
sage: QQ(113/40).partial_fraction_decomposition()
(2, [1/2, 1/8, 1/5])
sage: 2+1/2+1/8+1/5
113/40
```

Получим этот результат без привлечения готовой функции. Для этого разложим знаменатель на простые множители:

```
sage: 40.factor()
2^3 * 5
```

Таким образом,

$$p_1 = 2, \quad p_2 = 5.$$

Определим c_1 из уравнения

$$113 - c_1 p_2 = 0 \mod p_1.$$

В Sage это сделать не трудно:

```
sage: c1=GF(2)(113)/GF(2)(5)
sage: c1
1
```

Вычислим разность исходной дроби и простой дроби $c_1/2^2$:

```
sage: 113/40 - ZZ(c1)/2^3
27/10
```

Хорошо видно, что знаменатель потерял 2^2. Определим c_2 из уравнения

$$27 - c_2 p_2 = 0 \mod p_1,$$

используя:

```
sage: c2=GF(2)(27)/GF(2)(5)
sage: c2
1
sage: 27/10-ZZ(c2)/2
11/5
```

При этом разность дроби $27/10$ и $c_2/2$ равна

```
sage: 27/10-ZZ(c2)/2
11/5
```

Знаменатель потерял последнюю двойку. Поделим 11 на 5:

$$11 = 2 \cdot 5 + 1,$$

поэтому

$$\begin{aligned}\frac{113}{40} &= \frac{1}{2^3} + \frac{113}{40}\\ &= \frac{1}{2^3} + \frac{1}{2} + \frac{11}{5}\\ &= \frac{1}{2^3} + \frac{1}{2} + \frac{1}{5} + 2.\end{aligned}$$

Это и есть искомое разложение дроби $113/40$ на простые.

2. Поле частных кольца $k[x]$

Всякую дробь из поля частных кольца $k[x]$ можно представить как отношение двух многочленов. Числитель и знаменатель этой дроби можно разложить на простые множители. В этих разложения могут быть общие множители. В системах компьютерной алгебры всегда сокращают числитель и знаменатель на эти множители, поэтому мы далее будем считать, что числитель и знаменатель рассматриваемых дробей не имеют общих множителей.

Пример 2.

```
sage: FractionField(QQ[x])((x^2-1)/(x-1))
x + 1
```

Поскольку разложение многочлена на множители определено однозначно с точностью до мультипликативной константы, такое представление рациональной функции тоже определено с точностью до константы.

Определение 4. Дробь называется правильной, если степень ее числитель меньше степени знаменателя.

Определение 5. Правильная дробь называется простой (partial fraction), если ее знаменатель является степенью (power) простого многочлена, а степень числителя меньше степени (degree) этого простого многочлена.

Теорема 2. Всякую дробь из поля частных кольца $k[x]$ можно представить как сумму многочлена и нескольких простых дробей.

Определение 6. Представление дроби в виде сумму целого числа и нескольких простых дробей называется разложением на простые дроби (partial fraction dicomposition).

Пример 3.
```
sage: (QQ[x](x^15+1)/QQ[x]((x+1)^2*(x^2+2)^2)).partial_fraction_decomposition()
(x^9 - 2*x^8 - x^7 + 4*x^6 + 5*x^5 - 14*x^4 - 9*x^3 + 32*x^2 + 25*x - 82, [5/3/(x + 1), (-164/3*x + 569/3)/(x^2 + 2), (14*x - 57)/(x^4 + 4*x^2 + 4)])
sage: (RR[x](x^15+1)/QQ[x]((x+1)^2*(x^2+2)^2)).partial_fraction_decomposition()
(x^9 - 2.00000000000000*x^8 - x^7 + 4.00000000000000*x^6 + 5.00000000000000*x^5 - 14.0000000000000*x^4 - 9.00000000000000*x^3 + 32.0000000000000*x^2 + 25.0000000000000*x - 82.0000000000000, [1.66666666666667/(x + 1.00000000000000), 4.44089209850063e-16/(x^2 + 2.00000000000000*x + 1.00000000000000), (-54.6666666666668*x + 189.666666666667)/(x^2 +
```

```
    2.0000000000000), (14.000000000000*x -
    56.9999999999999)/(x^4 + 4.0000000000000*x^2 +
    4.0000000000000)])
sage: (CC[x](x^15+1)/QQ[x]((x+1)^2*(x^2+2)^2)).
    partial_fraction_decomposition()
(x^9 - 2.0000000000000*x^8 - x^7 +
    4.0000000000000*x^6 + 5.0000000000000*x^5 -
    14.000000000000*x^4 - 9.0000000000000*x^3 +
    32.000000000000*x^2 + 25.000000000000*x -
    82.000000000000, [(-27.3333333333334 -
    62.0191572665701*I)/(x - 1.41421356237310*I),
    (7.12500000000001 - 2.47487373415292*I)/(x^2 -
    2.82842712474619*I*x - 2.0000000000000),
    (-27.3333333333334 + 62.0191572665701*I)/(x +
    1.41421356237310*I), (7.12500000000001 +
    2.47487373415292*I)/(x^2 + 2.82842712474619*I*x -
    2.0000000000000), 1.66666666666667/(x +
    1.0000000000000), (4.44089209850063e-16)/(x^2 +
    2.0000000000000*x + 1.0000000000000)])
```

Разумеется, в двух последних случаях имеется ошибка округления. Разложения на множители знаменателя над $\mathbb{Q}$ и над $\mathbb{R}$ совпадают, поэтому и разложения на простые дроби тоже должны совпадать. Однако, из-за ошибки округления появляются малые числа (порядка 10^{-16}), которые на самом деле равны нулю.

Наиболее известным приложением разложения на простые дроби является вычисление интегралов от рациональных функций.

3. Интегрирование рациональных функций

Пусть дана рациональная функция $f = q/p$ с вещественными коэффициентами и требуется ее проинтегрировать. В поле частных кольца $\mathbb{R}[x]$ ее можно разложить на простые дроби, то есть представить в виде суммы элементарных дробей вида

$$\frac{b}{(x-a)^m}, \quad a, b \in \mathbb{R}, m \in \mathbb{N}$$

и вида

$$\frac{b_1 x + b_0}{(x^2 + a_1 x + a_0)^m}, \quad a_0, a_1, b_0, b_1 \in \mathbb{R}, m \in \mathbb{N},$$

а также некоторого многочлена. Поэтому вычисление интеграла сводится к интегрированию элементарных дробей.

Некоторые из них легко интегрируются. Прежде всего

$$\int \frac{bdx}{(x-a)^m} = -\frac{b}{(m-1)(x-a)^{m-1}}$$

при $m > 1$ и

$$\int \frac{bdx}{x-a} = b \ln |x-a|.$$

Трехчлен $x^2 + a_1 x + a_0$ должен быть простым над $\mathbb{R}$, поэтому его дискриминант $a_1^2 - 4a_0$ должен быть отрицательным. Выделим в трехчлене полный квадрат

$$x^2 + a_1 x + a_0 = \left(x + \frac{a_1}{2}\right)^2 + a_0 - \frac{a_1^2}{4}$$

и представим интеграл

$$\int \frac{(b_1 x + b_0)dx}{x^2 + a_1 x + a_0}$$

в виде сумма

$$b_1 \int \frac{\left(x + \frac{a_1}{2}\right) dx}{\left(x + \frac{a_1}{2}\right)^2 + a_0 - \frac{a_1^2}{4}} = \frac{b_1}{2} \ln \left(x^2 + a_1 x + a_0\right)$$

и

$$\int \frac{dx}{\left(x + \frac{a_1}{2}\right)^2 + a_0 - \frac{a_1^2}{4}} = \frac{2b_0 - a_1 b_1}{\sqrt{4\,a_0 - a_1^2}} \arctan\left(\frac{2x + a_1}{\sqrt{4\,a_0 - a_1^2}}\right).$$

Следует обратить внимание на то, что выражение, стоящее под логарифмом всегда положительно, равно как и выражение, стоящее под корнем.

Пример 4. Вычислим интеграл
$$\int \frac{x^5 dx}{(x+1)(x-2)(x-3)^2}.$$
Раскладываем на правильные дроби:

```
sage: (QQ[x](x^5)/QQ[x]((x+1)*(x-2)*(x-3)^2)).
   partial_fraction_decomposition()
(x + 7, [32/3/(x - 2), 1/48/(x + 1), 405/16/(x - 3),
    243/4/(x^2 - 6*x + 9)])
```

Это означает, что
$$\frac{x^5}{(x+1)(x-2)(x-3)^2} = x+7+\frac{32}{3}\frac{1}{x-2}+\frac{1}{48}\frac{1}{x+1}+\frac{405}{16}\frac{1}{x-3}+\frac{243}{4}\frac{1}{(x-3)^2}$$
и поэтому
$$\int \frac{x^5 dx}{(x+1)(x-2)(x-3)^2} = \frac{x^2}{2}+7x+\frac{32}{3}\ln|x-2|+\frac{1}{48}\ln|x+1|+ \\ +\frac{405}{16}\ln|x-3|-\frac{243}{4}\frac{1}{x-3}+C.$$

Вся трудность состоит в вычислении интегралов вида
$$\int \frac{(b_1 x+b_0)dx}{(x^2+a_1 x+a_0)^m}, \quad m>1,$$
возникающих в том случае, когда знаменатель f имеет кратные комплексные корни.

4. Интегрирование рациональных функций с целыми коэффициентами

Для функций с целыми коэффициентами Остроградский предложил весьма изящный способ обойти это затруднение, путем «перелета» из поля $\mathbb{R}$ в $\mathbb{C}$, а затем возвращения в $\mathbb{Q}$. Прежде всего поделим q на p, то есть запишем
$$f = u + \frac{r}{p},$$
Интегрирование многочлена u не представляет труда, поэтому ограничимся далее случаем, когда f — правильная дробь.

Теорема 3 (Остроградский)**.** Пусть g/h — правильная дробь с целыми коэффициентами, и разложение знаменателя h на простые множители в $\mathbb{Z}[x]$ имеет вид

$$h = p_1^{n_1} \ldots p_r^{n_r}.$$

Тогда

$$\frac{g}{h} = \frac{q_l}{p_1 \ldots p_r} + D\left(\frac{q_a}{p_1^{n_1-1} \ldots p_r^{n_r-1}}\right), \tag{1}$$

где $q_l, q_a \in \mathbb{Q}[x]$ и

$$\partial q_l < \partial p_1 + \cdots + \partial p_r, \quad \partial q_a < (n_1 - 1)\partial p_1 + \cdots + (n_r - 1)\partial p_r.$$

Доказательство. Над $\mathbb{C}$ простые многочлены — линейные, поэтому разложение $f = g/h$ на простые дроби представляет собой сумму элементарных дробей вида

$$\frac{b}{(x-a)^m}, \quad a, b \in \mathbb{C}.$$

Причем в эту сумму входят только такие дроби, у которых a — один из корней знаменателя p, а m не превышает кратность этого корня. Формула

$$\int \frac{bdx}{(x-a)^m} = -\frac{b}{(m-1)(x-a)^{m-1}}$$

при $m > 1$ остается в силе даже при комплексных a и b, а

$$\int \frac{bdx}{x-a}$$

мы выписать не станем, поскольку хотим избежать разговора о комплексном логарифме. Во всяком случае интеграл этого выражения не лежит в поле частных кольца $\mathbb{C}[x]$. Таким образом, разложение функции f на простые дроби можно разбить на две части: сумму дробей вида

$$\frac{b}{x-a},$$

именуемую логарифмической частью, и производную суммы дробей вида

$$\frac{b}{(x-a)^{m-1}},$$

где a — корень знаменателя p, а m не превышает его кратность. Эту сумму называют алгебраической частью интеграла.

Приведем слагаемые алгебраической части к общему знаменателю, в результате получим правильную дробь, знаменателем которой будет служить выражение

$$\prod (x - a_i)^{m_i - 1},$$

где $a_1, \ldots$ — комплексные корни h, а m_i — их кратности. Простые множители p_i не могут иметь общих комплексных корней или кратных корней. Поэтому корни h можно разбить на множество корней многочлена p_1 кратности n_1, множество корней многочлена p_2 кратности n_2 и т.д. Это означает, что

$$\prod (x - a_i)^{m_i - 1} = p_1^{n_1 - 1} \ldots p_r^{n_r - 1}.$$

Поэтому у алгебраической части мы знаем знаменатель и степень числителя, она строго меньше степени знаменателя.

Аналогично, логарифмическая часть представляет собой правильную дробь, знаменателем которой служит

$$\prod (x - a_i) = p_1 \ldots p_r.$$

Таким образом,

$$\frac{g}{h} = \frac{b_0 + b_1 x + \ldots b_N x^N}{p_1 \ldots p_r} + D\left(\frac{a_0 + a_1 x + \ldots a_M x^M}{p_1^{n_1 - 1} \ldots p_r^{n_r - 1}}\right),$$

где

$$N = \partial p_1 + \cdots + \partial p_r - 1, \quad M = (n_1 - 1)\partial p_1 + \cdots + (n_r - 1)\partial p_r - 1.$$

В этом соотношении нам пока не известны коэффициенты $a_0, \ldots, b_0, \ldots$, однако мы знаем, что их всегда можно подобрать в $\mathbb{C}$.

Это и есть формула (1), которую требовалось доказать. Остается доказать, что коэффициенты a_i, b_i принадлежит полю рациональных чисел. Применим для этого метод неопределенных коэффициентов. Введем символьные переменные $a_0, \ldots, b_0, \ldots$, вычислим производную и найдем числитель дроби

$$\frac{b_0 + b_1 x + \ldots b_N x^N}{p_1 \ldots p_r} + D\left(\frac{a_0 + a_1 x + \ldots a_M x^M}{p_1^{n_1 - 1} \ldots p_r^{n_r - 1}}\right) - \frac{g}{h}.$$

Этот числитель будет элементом $\mathbb{Z}[a_0, \ldots, b_0, \ldots][x]$. Коэффициенты этого многочлена будут линейными функциями $a_0, \ldots, b_0, \ldots$. Приравняв их нулю, мы получим систему линейных уравнений с целыми коэффициентами, подстановка решения которой превращает дробь в нуль.

Мы заранее знаем, что эта система с целыми коэффициентами имеет решение в $\mathbb{C}$. Поэтому она имеет решение, которое принадлежит $\mathbb{Q}$ и его можно найти по методу Гаусса. Взяв его, получим

$$q_a = a_0 + a_1 x + \ldots a_M x^M \in \mathbb{Q}[x]$$

и

$$q_l = b_0 + b_1 x + \ldots b_N x^N \in \mathbb{Q}[x].$$

□

Представленное доказательство теоремы Остроградского конструктивно, оно позволяет по заданной правильной дроби g/h найти q_l и q_a путем решения системы линейных уравнений с целыми коэффициентами. Более того, нетрудно заметить, что решение этой системы линейных уравнений единственно, поскольку иначе интегралы отличались бы более чем на аддитивную константу. Поэтому наша реализация метода Гаусса решения систем линейных уравнений позволяет реализовать метод Остроградского в Sage почти так, как он изложен в доказательстве теоремы 1, см. алгоритм 1.

Соотношение (1) позволяет свести интегрирование функции с целыми коэффициентами к интегрированию логарифмической части, то есть правильной дроби, знаменатель которой имеет только простые нули в $\mathbb{C}$. Для ее интегрирования мы воспользуемся разложением на простые дроби в

Algorithm 1 Отыскание представления правильной дроби в виде (1)

```
def ostrogradski(f,x):
    g=SR(f).numerator()
    h=SR(f).denominator()
    # Раскладываем знаменатель на множители
    L=list(ZZ[x](h).factor())

    # Составляем знаменатель алгебраической части
    hh=prod([p^(m-1) for (p,m) in L])
    n=hh.degree()
    if n==0:
        return [0,f]
    else:
        # Составляем числитель алгебраической части
        A=var(['A'+str(i) for i in range(n)])
        gg=sum([A[i]*x^i for i in range(n)])

        # Составляем знаменатель log части
        h3=ZZ[x](prod([p for (p,m) in L]))
         # Составляем числитель log части
        m=h3.degree()
        B=var(['B'+str(i) for i in range(m)])
        g3=sum([B[i]*x^i for i in range(m)])

        # Составляем выражение, которое должно быть равно нулю
        F=ZZ[A+B][x]((diff(SR(gg/hh),x)+SR(g3/h3)-f).numerator())
        # Работаем со СЛАУ
        S=tsolve(triangulation([QQ[A+B](eq) for eq in F.coefficients()]))
    # Список: алгебраическая часть, log часть
    return [(gg).subs(S)/hh, (g3).subs(S)/h3]
```

Algorithm 2 Интегрирование логарифмической части

```
def radical(a,r):
    if r==True:
        return AA(a).radical_expression()
    else:
        return AA(a)

def pfdintegral(f,x,r):
    g=SR(f).numerator()
    h=SR(f).denominator()

    if AA[x](h).degree()==1:
        return radical(g,r)*ln(abs(x+radical(h.subs(x=0),r)))
    else:
        b0=radical(g.subs(x=0),r)
        b1=radical(diff(g,x),r)
        a0=radical(h.subs(x=0),r)
        a1=radical(diff(h,x).subs(x=0),r)
        s=radical(sqrt(-a1^2 + 4*a0),r)
        return 1/2*b1*log(a1*x + x^2 + a0) - (a1*b1 - 2*b0)*arctan((a1 + 2*x)/s)/
```

Algorithm 3 Интегрирование рациональной функции с целыми коэффициентами

```
def rac_integral(f,x, list=False, radical=True):
    # Шаг 1. Приведение к правильной дроби
    g=SR(f).numerator()
    h=SR(f).denominator()
    [U,r]=QQ[x](g).quo_rem(QQ[x](h))
    # Шаг 2. Отыскание алгебраической части
    [A,L]=ostrogradski(SR(r/h),x)
    # Шаг 3. Интегрирование log части
    pfd=FractionField(AA[x])(L).partial_fraction_decomposition()[1]
    # Сборка ответа: интеграл многочлена + алгебраическая часть + интеграл от ло
    ans= [integral(U,x), A]+[pfdintegral(i,x,radical) for i in pfd]
    if list==True:
        return ans
    else:
        return sum(ans)
```

$\mathbb{A}[x]$, для чего, вообще говоря, придется решать нелинейные алгебраические уравнения, и далее вычислим интегралы по выписанным выше явным формулам. Эти формулы явно прописаны в алгоритме 2. При этом функция старается выразить коэффициенты в радикалах.

В результате интегрирование рациональной функции можно выполнить в три этапа:

1) приведение к правильной дроби при помощи стандартной функции `ruo_rem`,

2) выделение алгебраической и логарифмической частей по методу Остроградского, алгоритм 1,

3) интегрирование логарифмической части, алгоритм 2.

Все три шага собраны в алгоритме 3. Опция `list` позволяет записать ответ в виде списка из многочлена, алгебраической части и интеграла от логарифмической части.

Пример 5. Вычислим интеграл

$$\int \frac{x^4dx}{(x^3+8)^3}$$

по методу Остроградского:

```
sage: f=x^4/(x^3+8)^3
sage: rac_integral(f,x, list=True)
[0, 1/72*(x^5 - 4*x^2)/(x^6 + 16*x^3 + 64), -1/432*
   log(abs(x + 2)), 1/432*sqrt(3)*arctan(1/3*sqrt(3)
   *(x - 1)) + 1/864*log(x^2 - 2*x + 4)]
```

Дробь — правильная, поэтому первого слагаемого — многочлена — нет. Алгебраическая часть не содержит радикалов, как и утверждает теорема 3, а интеграл логарифмической части — содержит: в арктангенсе фигурирует $\sqrt{3}$. Этот интеграл может вычислить стандартная функция для вычисления интегралов в Sage:



Рис. 1. Вычисление интеграла в Wolfram Alpha.

```
sage: integral(f,x)
1/432*sqrt(3)*arctan(1/3*sqrt(3)*(x - 1)) + 1/72*(x
   ^5 - 4*x^2)/(x^6 + 16*x^3 + 64) + 1/864*log(x^2 -
   2*x + 4) - 1/432*log(x + 2)
```

Хорошо видно, что потерялся модуль у логарифма, поскольку система не может решить, работает ли она над $\mathbb{R}$ или над $\mathbb{C}$.

Пример 6. Вычислим интеграл

$$\int \frac{(3x^2+1)dx}{(2x^3+x^2+x+1)^2 x^3}$$

по методу Остроградского:

```
sage: f=(3*x^2 + 1)/(2*x^3+x^2+x+1)^2/x^3
sage: rac_integral(f,x, list=True, radical=False)
[0, 1/166*(852*x^4 + 40*x^3 + 441*x^2 + 249*x - 83)
  /(2*x^5 + x^4 + x^3 + x^2), 4*log(abs(x)),
  -4.034152213307781?*log(abs(x +
  0.7389836215045062?)), -0.514832231288379?*arctan
  (1.228750965490422?*x - 0.1468256778300298?) +
  0.01707610665389 02?*log(x^2 - 0.2389836215045063?*
  x + 0.6766049820996622?)]
```

Алгебраическая часть не содержит радикалов, как и утверждает теорема 3, а интеграл логарифмической части — содержит. Мы можем выразить алгебраические числа в радикалах, но выражение будет содержать мнимую единицу:

```
sage: rac_integral(f,x, list=True)
[0, 1/166*(852*x^4 + 40*x^3 + 441*x^2 + 249*x - 83)
  /(2*x^5 + x^4 + x^3 + x^2), 4*log(abs(x)),
  -1/498*((1/6889)^(1/3)*(5502141033*sqrt(249) -
  271425924878)^(1/3)*(-I*sqrt(3) + 1) +
  769473163*(1/6889)^(2/3)*(I*sqrt(3) + 1)
  /(5502141033*sqrt(249) - 271425924878)^(1/3) +
  664)*log(abs(x + 1/6*(3*sqrt(83)*sqrt(3) + 46)
  ^(1/3) - 5/6/(3*sqrt(83)*sqrt(3) + 46)^(1/3) +
  1/6)), 1/498*((1/6889)^(1/3)*(5502141033*sqrt(249)
   + 271425924878)^(1/3) + 769473163*(1/6889)^(2/3)
  /(5502141033*sqrt(249) + 271425924878)^(1/3) -
  664)*log(x^2 + 1/6*x*((3*sqrt(83)*sqrt(3) - 46)
  ^(1/3) - 5/(3*sqrt(83)*sqrt(3) - 46)^(1/3) + 2) +
  1/6*(1/2)^(1/3)*(3*sqrt(83)*sqrt(3) + 47)^(1/3) -
  2/3*(1/2)^(2/3)/(3*sqrt(83)*sqrt(3) + 47)^(1/3) +
```

```
1/6) - 1/249*(1/27)^(1/6)*(((1/6889)^(1/3)
*(5502141033*sqrt(249) + 271425924878)^(1/3) +
769473163*(1/6889)^(2/3)/(5502141033*sqrt(249) +
271425924878)^(1/3) - 664)*((3*sqrt(83)*sqrt(3) -
46)^(1/3) - 5/(3*sqrt(83)*sqrt(3) - 46)^(1/3) + 2)
 - 6*(1/6889)^(1/3)*(465178335771*sqrt(249) -
3164433545876)^(1/3) + 40264122198*(1/6889)^(2/3)
/(465178335771*sqrt(249) - 3164433545876)^(1/3) -
564)*(276*sqrt(83)*sqrt(3) + 4357)^(1/6)*arctan
((1/27)^(1/6)*(276*sqrt(83)*sqrt(3) + 4357)^(1/6)
*(12*x + (3*sqrt(83)*sqrt(3) - 46)^(1/3) - 5/(3*
sqrt(83)*sqrt(3) - 46)^(1/3) + 2)/((276*sqrt(83)*
sqrt(3) + 4357)^(1/3) + 5))/((276*sqrt(83)*sqrt(3)
 + 4357)^(1/3) + 5)]
```

Стандартная функция для вычисления интегралов в Sage найти этот интеграл не может:

```
sage: integral(f,x)
1/166*(852*x^4 + 40*x^3 + 441*x^2 + 249*x - 83)/(2*x
   ^5 + x^4 + x^3 + x^2) - 1/83*integrate((664*x^2 -
   94*x + 505)/(2*x^3 + x^2 + x + 1), x) + 4*log(x)
```

Коммерческие системы компьютерной алгебры, напр., Wolfram Alpha, предлагают ответ без арктангенса, то есть через комплексные логарифмы, см. рис. 1.

5. Задания

Теоретические вопросы.

1) Дайте определение простой дроби в $\mathbb{Q}$ и в поле частных кольца $\mathbb{Q}[x]$.

2) Докажите теорему 2.

3) Сформулируйте теорему Остроградского. Как она используется для вычисления интегралов?

Практические задания.

1) Разложите дробь 14515/39168 на простые дроби.

2) Разложите дробь

$$\frac{x^9}{(x^2-2)(x-3)^3(x^2+1)}$$

на простые дроби в полях частных колец a.) $\mathbb{Q}[x]$, b.) $\mathbb{R}[x]$, c.) $\mathbb{A}[x]$, d.) $\mathbb{C}[x]$, e.) $\overline{\mathbb{Q}}[x]$.

3) Найдите интеграл

$$a.) \quad \int \frac{xdx}{(x-1)(x-2)^2(x-3)^3}$$

$$b.) \quad \int \frac{dx}{x^3+1}$$

$$c.) \quad \int \frac{xdx}{(x^3+1)^5}$$

$$d.) \quad \int \frac{x^4dx}{x^3+1}$$

$$e.) \quad \int \frac{dx}{(x^3+x+1)^3(x^2+1)^2}$$

$$h.) \quad \int \frac{(3x^2+1)dx}{(x^3+x+1)^3}$$

При возможности выразите все константы в радикалах.